#include          "msp430x14x.h" 

;******************************************************************************

;   MSP-FET430P140 - Power Up sequencing of voltage rails.

;         

;       Control Method = VOLTAGE

;

;       Name: Seq_rails_ADC

;

;   Description: 

;   

;  

;   This program will provide control and monitoring signals for the powering up and down 

;   of 3 voltage rails based on voltage control.

;   The power supplies are three low dropout regulators, LDO's, TPS725xx, which have active 

;   HIGH enables. 

;   P1.1, P1.2 ,P1.3 provide the enable signals.

;   P6.0(AD0), P6.1(AD1), P6.2(AD2) are ADC inputs for monitoring the output voltage of each LDO. 

;   P1.0  is the shutdown input

;   P1.4 provides a system Reset

;

;   

;  SMCLK = MCLK = TACLK = default DCO ~ 800kHz, therefore  base tick is appro. 1.25usec. Exact time not critical.

;  

;  

;                  

;

;     

;                MSP430F149

;              ---------------

;             |               |

;             |      TA0(P1.1)|13----> Control Regulator #3  3.3V

;             |               |

;             |      TA1(P1.2)|14----> Control Regulator #1  1.8 V 

;             |               |

;             |      TA2(P1.3)|15----> Control Regulator #2  2.5V

;             |               |

;             |          P6.0 |59----< Output Regulator #2

;             |               |

;             |          P6.1 |60----< Output Regulator #1

;             |               |

;             |          P6.2 |61----< Output Regulator #3

;             |               |

;             |          P1.4 |16---> System RESET

;             |               |

;             |          P1.0 |12---< Shutdown

;

;

;       Joe Di Bartolomeo 

;       Texas Instruments, Inc

;       July , 2002

;******************************************************************************

;------------------------------------------------------------------------------ 

main        ORG     01100h                  ; Program Start

;------------------------------------------------------------------------------ 

RESET       mov     #0A00h,SP               ; Initialize stackpointer

StopWDT     mov     #WDTPW+WDTHOLD,&WDTCTL  ; Stop watchdog

            mov.b   #007h,&P6SEL            ; Assign ADC -P6.0(AD0), P6.1(AD1), P6.2(AD2)

SetupP1     bis.b   #01Eh,&P1DIR            ; Port 1 OUTPUTS( 1,2,3,4) INPUTS(0,5,6,7)

            mov.b   #000h, &P1OUT           ; Insure P1.1,P1.2 and P1.3 are LOW  on start up. 

                                            ; SYSTEM RESET P1.4 LOW  

            mov.w   #0980h,R10              ; ADC value for 1.8V

            mov.w   #0C00h,R11              ; ADC value for 2.5V

            mov.w   #0580h,R12              ; ADC value for 1.2 V, feedback on adjustable LDO for 3.3V                                                                          ;

SetupADC12 

            mov.w   #0410h,&ADC12CTL0       ; ADC12ON, S/H =04h 

            mov.w   #0210h,&ADC12CTL1       ; single sample, SHP=1,ADC12OSC, ADC12DIV=2

Setup_Rail1

            xor.b   #004h, &P1OUT           ; Turn on Rail #1                               

            mov.b   #00h,&ADC12MCTL0        ; Select A0, VCC and -Vss are Ref

            bis     #ENC,&ADC12CTL0         ; Enable conversions

                                            ;            

Adc_Rail1   bis     #ADC12SC,&ADC12CTL0     ; Start conversions

testIFG0    bit     #BIT0,&ADC12IFG         ; Conversion done?

            jz      testIFG0                ; No, test again

            cmp R10, &ADC12MEM0             ; Is rail #1 within spec?

            jn Adc_Rail1                    ; If not take another sample

;Rail # 1 is within spec.          

Setup_Rail2

            xor.b   #008h, &P1OUT           ; Turn on Rail #2                               

            bic     #ENC,&ADC12CTL0         ; Disable conversions

            mov.b   #01h,&ADC12MCTL0        ; Select A1

            bis     #ENC,&ADC12CTL0         ; Enable conversions

                                            ;            

Adc_Rail2   bis     #ADC12SC,&ADC12CTL0     ; Start conversions

testIFG1    bit     #BIT0,&ADC12IFG         ; Conversion done?

            jz      testIFG1                ; No, test again

            cmp R11, &ADC12MEM0             ; Is rail #2 within spec?

            jn Adc_Rail2                    ; If not take another sample

;Rail #2 is within spec.

Setup_Rail3

            xor.b   #002h, &P1OUT           ; Turn on Rail #3                             

            bic     #ENC,&ADC12CTL0         ; Disable conversions

            mov.b   #02h,&ADC12MCTL0        ; Select A2

            bis     #ENC,&ADC12CTL0         ; Enable conversions

                                            ;            

Adc_Rail3  bis     #ADC12SC,&ADC12CTL0      ; Start conversions

testIFG2    bit     #BIT0,&ADC12IFG         ; Conversion done?

            jz      testIFG2                ; No, test again

            cmp R12, &ADC12MEM0             ; Is rail #3 within spec?

            jn Adc_Rail3                    ; If not take another sample

;At this point all rails are up and within tolerance. Apply System Reset

            mov.w  #4,R14                   ;

L3                                          ; This delay is the System RESET delay

            mov.w   #10000,R15              ; User defined               

L4          dec.w   R15                    

            jnz     L4                     

            dec.w   R14

            jnz     L3

            xor.b   #010h, &P1OUT           ; Take System Reset High, SYSTEM RESET.

Monitor                                     ; Monitor LDO voltage. Insure voltages are allows within spec.

            bic      #ENC,&ADC12CTL0        ; Disable conversions

            mov.b    #00h,&ADC12MCTL0       ; Select A0

            mov.b    #01h,&ADC12MCTL1       ; Select A1

            mov.b    #82h,&ADC12MCTL2       ; Select A2, and set EOC

            mov.w    #MSH+ADC12ON,&ADC12CTL0

            mov.w    #SHP+CONSEQ_1,&ADC12CTL1

            bis      #ENC,&ADC12CTL0        ; Enable conversions

Keep_Mon            

            bis     #ADC12SC,&ADC12CTL0     ; Start conversions

Mon_Test    bit     #BIT2,&ADC12IFG         ; Conversion done?

            jz      Mon_Test                ; No, test again

            cmp R10,&ADC12MEM0              ; Test rail #1

            jn Error                        ; If out of spec jump to Error routine   

            cmp R11,&ADC12MEM1              ; Test rail #2

            jn Error                        ; If out of spec jump to Error routine 

            cmp R12,&ADC12MEM2              ; Test rail #3

            jn Error                        ; If out of spec jump to Error routine

            bit.b   #01h,&P1IN              ; A low on this input signals the MSP430 to powerdown.

            jnc    Powerdown                ; Normal operation should be High, If Low enter power down.

            br    #Keep_Mon

Error                                       ; An error handler would be placed here which would be                

          nop                               ; application dependant

Powerdown

                                            ; Power Down is user defined. Here a simple power down is

                                            ; performed.                                      

            xor.b   #010h, &P1OUT           ; Take System Reset low, (optional)

            xor.b   #002h, &P1OUT           ; Turn off Rail #3                   

            xor.b   #008h, &P1OUT           ; Turn off Rail #2    

            xor.b   #004h, &P1OUT           ; Turn off Rail #1    

here        nop

            jmp     here                    ; 

                                            ;

;------------------------------------------------------------------------------ 

;           Interrupt Vectors        

;------------------------------------------------------------------------------

            ORG     0FFFEh                  ; MSP430 RESET Vector

            DW      RESET                   ;

            END

